Computing New Variables in SAS
The SAS Data Step is a powerful and flexible programming tool that is used to create a new SAS dataset. A Data Step is required to create any new variables or modify existing ones in SAS. The Data Step allows you to assign a particular value to all cases, or to a subset of cases; to transform a variable by using a mathematical function, such as the log function; or to create a sum, average, or other summary statistic based on the values of several existing variables within an observation.

Unlike Stata and SPSS, you cannot simply create a new variable or modify an existing variable in “open” SAS code. This basically means that you need to create a new dataset by using a Data Step whenever you want to create or modify variables. A single Data Step can be used to create an unlimited number of new variables. 

We will illustrate creating new variables using the employee.sas7bdat SAS dataset, which has 474 observations and 10 variables. We assume this dataset has been downloaded and saved in the C:\ folder. 

To create the new dataset called “mylib.employee2”, first highlight and submit the Libname statement. The Libname statement needs to be submitted only once per SAS session. It tells SAS where to read and write permanent SAS datasets. 

The Data Step starts with the Data Statement and ends with Run. You create the new dataset called mylib.employee2 by highlighting and submitting the entire Data Step commands shown below, starting at “data” and ending with “run;”.  Each time you make any changes to the Data Step commands, you must highlight and re-submit them. This will re-create the mylib.employee2 dataset by over-writing the previous version. 

libname mylib "C:\";

data mylib.employee2;

  set mylib.employee;
    /* put commands to create new variables here*/

    /* be sure they go BEFORE the run statement*/
run;
We discuss each aspect of the SAS Data Step statements below. 

Libname statement:

libname mylib "C:\";

The Libname statement tells SAS the folder where existing SAS datasets are located, and where new permanent SAS datasets can be saved. In this case we assign “MYLIB” to the folder C:\. The general form of the Libname statement is:  libname libref "path_to_folder"; The folder must already exist before it can be used in a Libname statement.

You can specify the type of dataset that you wish to read/write to a folder by adding a SAS “Engine” to the Libname statement. To tell SAS to read or create Version 9 SAS datasets in your folder, you could use syntax like the following:  libname libref V9 "path_to_folder\"; If you’re using SAS release 9, the default is to create version 9 datasets.
Data Statement:

data mylib.employee2;

The Data statement tells SAS to create a new permanent SAS dataset called mylib.employee2. The first name of the two-level dataset name is the library (mylib) and the second part of the name follows a period (.) and is the name of the dataset itself (employee2). In Windows, this dataset will be named employee2.sas7bdat, however the file extension is not used in the SAS code. In general, if you want to create a new permanent SAS dataset, you would use SAS code of the form: data library.datasetname; 
Set Statement:

set mylib.employee;
The Set statement tells SAS to create your new dataset from the existing permanent SAS dataset called mylib.employee, which in Windows is the employee.sas7bdat within the C:\ folder.

Run Statement:

run;
The Run statement is a “Step Boundary” in SAS. It ends the SAS Data Step, Remember to include all codes to create new variables before the Run statement. 

The example below illustrates creating a number of new variables in the new dataset, mylib.employee2, which is created from mylib.employee. 

Examples of computing new variables SAS:

libname mylib "C:\";

data mylib.employee2;

  set mylib.employee;

/*Generating new variables containing constants*/

  currentyear=2005;

  alpha ="A"; 

  may17 = "17MAY2000"D;
  format may17 mmddyy10.;

  format bdate mmddyy10.;
/* Generating Variables Using Values from Other Variables*/
  saldiff = salary - salbegin;
  label saldiff = "Current Salary – Beginning Salary";
/* Generating Variables Conditionally Based on Values of Other Variables */
  if (salary >= 0 & salary <= 25000) then salcat = "C"; 
  if (salary > 25000 & salary <= 50000) then salcat = "B";
  if (salary > 50000) then salcat = "A"; 
  if salary not=. and jobcat not=. then do;

    if (salary < 50000 & jobcat = 3) then manlowsal = "Y"; 
    else manlowsal = "N";

  end;
  if (salary >= 0 & salary <= 25000) then nsalcat = 1; 
  if (salary > 25000 & salary <= 50000) then nsalcat = 2;
  if (salary > 50000) then nsalcat = 3; 
/*Generating dummy variables*/
minhighsal = (salary > 40000 & minority = 1);
if (salary not=. and minority not=.) then minhighsal = (salary > 40000 & minority = 1);
if jobcat not=. then do;

   jobdum1 = (jobcat=1);

   jobdum2 = (jobcat=2);

   jobdum3 = (jobcat=3);

end;

/*Using statistical functions*/

nmiss = nmiss(of bdate--minority);
npresent = n(of bdate--minority);
salmean = mean(salary, salbegin);
run;
There are many functions that can be used in SAS to create new variables. The following list contains some of the more common SAS operators and functions:

Arithmetic Operators:

+
Addition

-
Subtraction

*
Multiplication
/
Division

**
Exponentiation

Arithmetic Functions:

ABS

Absolute value

ROUND(arg,unit) 
Rounds argument










to the nearest unit 

INT

Truncate


MOD(arg,divisor)
Modulus 










(remainder)

SQRT

Square root


EXP


Exponential

LOG10
Log base 10


LOG


Natural log

ARSIN
Arcsin


ATAN


Arctangent

SIN

Sine



COS


Cosine

Statistical Functions (Arguments can be numeric values or variables):

SUM(Arg1, Arg2,…,ArgN)


Sum of non-missing arguments

MEAN(Arg1, Arg2,…,ArgN)

Mean of non-missing arguments

STD(Arg1, Arg2,…,ArgN)


Standard deviation of non-missing 







arguments

VAR(Arg1, Arg2,…,ArgN)


Variance of non-missing arguments

CV(Arg1, Arg2,…,ArgN)
Coefficient of variation of non-missing arguments

MIN(Arg1, Arg2,…,ArgN)


Minimum of non-missing arguments

MAX(Arg1, Arg2,…,ArgN)


Maximum of non-missing arguments

Missing Values Functions:
MISSING(Arg)
= 1 if the value of Arg is missing         
= 0 if not missing

NMISS(Var1, Var2,…,VarN)
Number of missing values across variables within a case

N(Var1, Var2,…,VarN)
Number of non-missing values across variables within a case

Across-case Functions:
LAG(Var)




Value from previous case

LAGn(Var)




Value from nth previous case

Date and Time Functions:

Datepart(datetimevalue)
Extracts date portion from a datetime value

Month(datevalue)



Extracts month from a date value

Day(datevalue)



Extracts day from a date value

Year(datevalue)



Extracts year form a date value
Intck(‘interval’,datestart,dateend)
Finds the number of completed intervals between two dates
Other Functions:

RANUNI(Seed)
Uniform pseudo-random no. defined on the interval (0,1)
RANNOR(Seed)



Std. Normal pseudo-random no.

PROBNORM(x)



Prob. a std. normal  is <= x

PROBIT(p) 




pth quantile from std. normal dist.

Numeric vs. Character Variables

There are only two types of variable in SAS: numeric and character. Numeric variables are the default type and are used for numeric and date values. 

Character variables can have alpha-numeric values, which may be any combination of letters, numbers, or other characters. The length of a character variable can be up to 32767 characters. Values of character variables are case-sensitive. For example, the value “Ann Arbor” is different than the value “ANN ARBOR”. 
A Quick Note about variable names in SAS

Variable names in SAS can be up to 32 characters long, and must start with either a letter or underscore. Variable names may contain only letters, numbers, and underscores. No blanks are allowed. Variable names will be saved in a SAS dataset in the combination of upper and lower case used when defining the variable (i.e. when the variable name is first mentioned in a Data Step). Although SAS will remember the variable name using the case you specify when creating it, it will recognize the variable name if you type it using any case later on.

Generating Variables Containing Constants

We first consider the simple case in which we want to create a variable that has a constant value for all observations in the data set. In the example below we create a new numeric variable named “currentyear”, which has a constant value of 2005 for all observations:

currentyear=2005;
The example below illustrates creating a new character variable named “alpha” which contains the letter “A” for all observations in the dataset. Note that the value must be enclosed either in single or double-quotes, because this is a character variable. 

alpha ="A"; 

Dates can be generated in a number of different ways. For example, we can use the mdy function to create a date value from a month, day, and year value, as shown below:

may17 = mdy(5,17,2000); 
Or we can create a date by using a SAS date constant, as shown below:

may17 = "17MAY2000"D; 
The D following the quoted date constant tells SAS that this is not a character variable, but a date value, which is stored as a numeric value.
If we now look at the value of may17 by typing “viewtable mylib.employee2” or “vt mylib.employee2” in the SAS command dialog box in the upper left of the SAS desktop, we will see that the variable “may17” contains the value 14747 for each observation. This is because SAS stores dates as the number of days since January 1, 1960. We can reformat this variable so that it actually displays as a date by typing the following statement in our Data Step, and resubmitting it to recreate the dataset. 

format may17 mmddyy10.;
NB: Remember to close the viewtable window showing the data before resubmitting the commands, because SAS locks a dataset when it is open in the viewtable window, and will not allow any changes to be made while the viewtable window is open.

If you once again open the dataset in the viewtable window, you will now see it displayed as 05/17/2000. Remember to close the viewtable window when you are done viewing the dataset.

Generating Variables Using Values from Other Variables

We can also generate new variables as a function of existing variables. For example, if we wanted to create a new variable in the mylib.employee2 data set containing the difference between the current salary and beginning salary for each employee, we could simply use the following statement:

saldiff = salary – salbegin;
New variables can also be labeled:

label saldiff = "Current Salary – Beginning Salary";
We can use the mdy function to create a new date value, based on three variables in a dataset, in this example the variables were called “Month”, “Day”, and “Year”, although they could have different names:

date = mdy(month,day,year);
Values of the date variable would vary from observation to observation, because the mdy() function is using different values of variables to create date. Remember to use a Format statement to format the new variable DATE so it will look like a date.

Generating Variables Conditionally Based on Values of Other Variables

You can also create new variables in SAS conditional on the values of other variables. For example, if we wanted to create a new character variable, SALCAT, that contains salary categories “A”, “B”, and “C” we could use the following commands. Note that we have used conditional syntax to give new values to all possible original values of salary. If we had accidentally left out any categories of salary, they would be given a missing value in SALCAT.

if (salary >= 0 & salary <= 25000) then salcat = "C"; 
if (salary > 25000 & salary <= 50000) then salcat = "B";
if (salary > 50000) then salcat = "A"; 
Note the use of an If…then statement to identify the condition that a given case in the data set must meet for the new variable to be given a value of “A”. In general, these types of conditional commands have the form:

if (condition) then varname = value;

where the condition can be specified using a logical operator or a mnemonic (e.g., =  (eq), & (and), | (or), ~= (not=, ne), > (gt), >= (ge) < (lt) <= (le)). The parentheses are not necessary to specify a condition in SAS, but can be used to clarify a statement or to group parts of a statement. A semicolon is required at the end of the statement. For example, if one wants to create a variable that identifies employees who are managers but have relatively low salaries, one could use a statement like

if (salary < 50000 & jobcat = 3) then manlowsal = "Y";
This will create a new character variable equal to “Y” whenever an employee meets those conditions on the two variables, salary and jobcat. However, this variable may be incorrectly coded, due to the presence of missing values, as discussed in the note below. 

Note on missing values when conditionally computing new variables in SAS:
When computing new values conditionally based on the values of other variables, SAS users need to be careful with missing data. SAS considers missing values for numeric variables to be smaller than the smallest possible numeric value in a data set. Therefore, in the salary condition above, if an employee had missing data on the salary variable, that employee would be coded into category “Y” on the new MANLOWSAL variable. A safer version of this conditional command would look like this:

if (salary not=. & salary < 50000 & jobcat = 3) then manlowsal = "Y"; 

The condition now emphasizes that salary must be less than $50,000 and not equal to a missing value. 

The following statements could be used to set up a variable with a value of “Y” or “N” on the new variable MANLOWSAL. Note that the use of ‘else’ will put all values, including missing values on either variable, into the “N” category (every other value, including missing, is captured by the ‘else’ condition). The final If statement will put anyone with a missing value on either of these variables into the missing value of MANLOWSAL, which is

 " " for a character variable. 

if (salary not=. & salary < 50000 & jobcat = 3) then manlowsal = "Y"; 
else manlowsal = "N";

if salary = . or jobcat=. then manlowsal=" ";

Another way this could be done would be to use a Do Loop before creating the variable, as shown below. If you use a do; statement, you must have an end; statement to close the do loop. In the example below, the entire block of code will only be executed if salary is not missing and jobcat is not missing.

if salary not=. and jobcat not=. then do;

  if (salary < 50000 & jobcat = 3) then manlowsal = "Y"; 
  else manlowsal = "N";

end;
SAS users need to keep the handling of missing values in mind when using conditional recodes like this.
If you had wanted to create new numeric variables conditionally, you could use syntax like that shown below. Note that quotations are not put around the values of numeric variables.

if (salary >= 0 & salary <= 25000) then nsalcat = 1; 
if (salary > 25000 & salary <= 50000) then nsalcat = 2;
if (salary > 50000) then nsalcat = 3; 
It is often easier to deal with numeric values than character values in SAS.

Generating Dummy Variables

Statistical analyses often require SAS users to generate dummy variables, which are also known as indicator variables, Dummy variables take on a value of 1 for certain cases, and 0 for all other cases. A common example is the creation of a dummy variable to represent gender, where the value of 1 might identify females, and 0 males. 

The following statements can be used to create a dummy variable that indicates employees with salaries over $40,000 and minority status:

minhighsal = (salary > 40000 & minority = 1);
Take a look at the data set in the viewtable window (type “viewtable mylib.employee2” or “vt mylib.employee2” in the command dialog box in the upper right-hand corner of the SAS desktop). How many of the 474 employees have a value of 1 on this variable? 

SAS has automatically generated a 1 for all cases meeting the condition specified in the parentheses, and a 0 for all cases not meeting the condition (including cases with missing values on either variable). To be sure that no missing values got mistakenly included in the minhighsal variable, you could use the following syntax:

if (salary not=. and minority not=.) then minhighsal = (salary > 40000 & minority = 1);
If you have a variable with more than 3 or more categories, you can create a dummy variable for each category, and later in a later analysis, you would usually choose to include one less dummy variable in a model than there are categories. For example, in the employee dataset, there is a variable called jobcat, with 3 levels (1, 2, and 3). Here is SAS code to create three new dummy variables, one for each level of jobcat. Notice that potential problems with missing values are avoided by using a do loop prior to the creation of the dummy variables:

if jobcat not=. then do;

   jobdum1 = (jobcat=1);

   jobdum2 = (jobcat=2);

   jobdum3 = (jobcat=3);

end;

SAS will not give you an error message if you generate a new variable with the same name as a previously existing variable. It simply replaces the previous values of the variable with the new values. 
You can apply value labels (called user-defined formats in SAS) to a new variable once you have generated it; this will be described later. 

Using Statistical Functions

You can also use SAS to determine how many missing values are present in a list of variables within each observation, as shown in the example below:

nmiss = nmiss(of bdate--minority);
Note the use of the double dashes (--) to indicate a variable list (with variables given in dataset order). Be sure to use “of” when using a variable list like this.

The converse operation is to determine the number of non-missing values there are in a list of variables, 

npresent = n(of bdate--minority);
Another common operation is to calculate the sum or the mean of the values for several variables and store the results in a new variable. For example, to calculate a new variable, salmean, representing the average of the current and beginning salary, use the following command. Note that you can use a list of variables separated by commas, without including “of” before the list.

salmean = mean(salary, salbegin);
All missing values for the variables listed will be ignored when computing the mean in this way. The min( ), max( ), and std( ) functions work in a similar way. 

